

## COMP1022Q Recursion Review Question

Here is some VBA code.

```
Option Explicit

Dim MaxDepth As Integer

Sub InputMaxDepth()
    Dim CenterX As Double, CenterY As Double, Radius As Double

    CenterX = 250
    CenterY = 350
    Radius = 128

    MaxDepth = InputBox("Enter the maximum " & _
        "depth for drawing the circles: ")

    DrawCircle CenterX, CenterY, Radius, 0
End Sub

Sub DrawCircle(CenterX As Double, CenterY As Double, _
    Radius As Double, Depth As Integer)

    Dim GrayColour As Double
    Dim Top As Double, Left As Double
    Dim Width As Double, Height As Double
    Dim CircleObj As Shape


    If MaxDepth > 0 Then
        GrayColour = 255 * (CDBl(MaxDepth - Depth) / (MaxDepth + 1))
    End If

    Left = CenterX - Radius
    Top = CenterY - Radius
    Width = Radius * 2
    Height = Radius * 2

    Set CircleObj = ActiveSheet.Shapes.AddShape _
        (msoShapeOval, Left, Top, Width, Height)
    CircleObj.Line.Visible = msoFalse
    CircleObj.Fill.ForeColor.RGB = _
        RGB(GrayColour, GrayColour, GrayColour)

    If Depth < MaxDepth Then
        DrawCircle CenterX + Radius, CenterY, Radius / 2, Depth + 1
        DrawCircle CenterX - Radius, CenterY, Radius / 2, Depth + 1
        DrawCircle CenterX, CenterY - Radius, Radius / 2, Depth + 1
        DrawCircle CenterX, CenterY + Radius, Radius / 2, Depth + 1
    End If
End Sub
```

When the `InputMaxDepth()` subroutine is executed the user is asked to enter a number (the maximum depth).



What number must the user enter to create the display shown above?

**Answer:** \_\_\_\_\_